

## F11DBF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F11DBF solves a system of linear equations involving the incomplete  $LU$  preconditioning matrix generated by F11DAF.

### 2 Specification

```

SUBROUTINE F11DBF(TRANS, N, A, LA, IROW, ICOL, IPIVP, IPIVQ, ISTR,
1          IDIAG, CHECK, Y, X, IFAIL)
INTEGER    N, LA, IROW(LA), ICOL(LA), IPIVP(N), IPIVQ(N),
1          ISTR(N+1), IDIAG(N), IFAIL
  real     A(LA), Y(N), X(N)
CHARACTER*1 TRANS, CHECK

```

### 3 Description

This routine solves a system of linear equations

$$Mx = y, \quad \text{or} \quad M^T x = y,$$

according to the value of the argument TRANS, where the matrix  $M = PLDUQ$ , corresponds to an incomplete  $LU$  decomposition of a sparse matrix stored in coordinate storage (CS) format (see Section 2.1.1 of the Chapter Introduction), as generated by F11DAF.

In the above decomposition  $L$  is a lower triangular sparse matrix with unit diagonal elements,  $D$  is a diagonal matrix,  $U$  is an upper triangular sparse matrix with unit diagonal elements and  $P$  and  $Q$  are permutation matrices.  $L$ ,  $D$  and  $U$  are supplied to F11DBF through the matrix

$$C = L + D^{-1} + U - 2I$$

which is an  $N$  by  $N$  sparse matrix, stored in CS format, as returned by F11DAF. The permutation matrices  $P$  and  $Q$  are returned from F11DAF via the arrays IPIVP and IPIVQ.

It is envisaged that a common use of F11DBF will be to carry out the preconditioning step required in the application of F11BBF to sparse linear systems. F11DBF is used for this purpose by the black-box routine F11DCF.

F11DBF may also be used in combination with F11DAF to solve a sparse system of linear equations directly (see Section 8.5 of the document for F11DAF). This use of F11DBF is demonstrated in Section 9.

### 4 References

None.

### 5 Parameters

1: TRANS — CHARACTER\*1 *Input*

*On entry:* specifies whether or not the matrix  $M$  is transposed:

- if TRANS = 'N', then  $Mx = y$  is solved;
- if TRANS = 'T', then  $M^T x = y$  is solved.

*Constraint:* TRANS = 'N' or 'T'.

- 2:** N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $M$ . This **must** be the same value as was supplied in the preceding call to F11DAF.  
*Constraint:*  $N \geq 1$ .
- 3:** A(LA) — *real* array *Input*  
*On entry:* the values returned in the array A by a previous call to F11DAF.
- 4:** LA — INTEGER *Input*  
*On entry:* the dimension of the arrays A, IROW and ICOL as declared in the (sub)program from which F11DBF is called. This **must** be the same value as was supplied in the preceding call to F11DAF.
- 5:** IROW(LA) — INTEGER array *Input*
- 6:** ICOL(LA) — INTEGER array *Input*
- 7:** IPIVP(N) — INTEGER array *Input*
- 8:** IPIVQ(N) — INTEGER array *Input*
- 9:** ISTR(N+1) — INTEGER array *Input*
- 10:** IDIAG(N) — INTEGER array *Input*  
*On entry:* the values returned in arrays IROW, ICOL, IPIVP, IPIVQ, ISTR and IDIAG by a previous call to F11DAF.
- 11:** CHECK — CHARACTER\*1 *Input*  
*On entry:* specifies whether or not the CS representation of the matrix  $M$  should be checked:  
     if CHECK = 'C', checks are carried on the values of N, IROW, ICOL, IPIVP, IPIVQ, ISTR and IDIAG;  
     if CHECK = 'N', none of these checks are carried out.  
 See also Section 8.2.  
*Constraint:* CHECK = 'C' or 'N'.
- 12:** Y(N) — *real* array *Input*  
*On entry:* the right-hand side vector  $y$ .
- 13:** X(N) — *real* array *Output*  
*On exit:* the solution vector  $x$ .
- 14:** IFAIL — INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

## 6 Errors and Warnings

If on entry `IFAIL = 0` or `-1`, explanatory error messages are output on the current error message unit (as defined by `X04AAF`).

Errors detected by the routine:

`IFAIL = 1`

On entry, `TRANS`  $\neq$  'N' or 'T',  
or `CHECK`  $\neq$  'C' or 'N'.

`IFAIL = 2`

On entry, `N`  $<$  1.

`IFAIL = 3`

On entry, the CS representation of the preconditioning matrix  $M$  is invalid. Further details are given in the error message. Check that the call to `F11DBF` has been preceded by a valid call to `F11DAF` and that the arrays `A`, `IROW`, `ICOL`, `IPIVP`, `IPIVQ`, `ISTR` and `IDIAG` have not been corrupted between the two calls.

## 7 Accuracy

If `TRANS = 'N'` the computed solution  $x$  is the exact solution of a perturbed system of equations  $(M + \delta M)x = y$ , where

$$|\delta M| \leq c(n)\epsilon P|L||D||U|Q,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*. An equivalent result holds when `TRANS = 'T'`.

## 8 Further Comments

### 8.1 Timing

The time taken for a call to `F11DBF` is proportional to the value of `NNZC` returned from `F11DAF`.

### 8.2 Use of CHECK

It is expected that a common use of `F11DBF` will be to carry out the preconditioning step required in the application of `F11BBF` to sparse linear systems. In this situation `F11DBF` is likely to be called many times with the same matrix  $M$ . In the interests of both reliability and efficiency, you are recommended to set `CHECK` to 'C' for the first of such calls, and to 'N' for all subsequent calls.

## 9 Example

This example program reads in a sparse nonsymmetric matrix  $A$  and a vector  $y$ . It then calls `F11DAF`, with `LFILL = -1` and `DTOL = 0.0`, to compute the **complete**  $LU$  decomposition

$$A = PLDUQ.$$

Finally it calls `F11DBF` to solve the system

$$PLDUQx = y.$$

## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*   F11DBF Example Program Text
*   Mark 18 Release. NAG Copyright 1997.
*   .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          NMAX, LA, LIWORK
PARAMETER        (NMAX=1000,LA=10000,LIWORK=7*NMAX+2)
*   .. Local Scalars ..
real           DTOL
INTEGER          I, IFAIL, LFILL, N, NNZ, NNZC, NPIVM
CHARACTER        CHECK, MILU, PSTRAT, TRANS
*   .. Local Arrays ..
real           A(LA), X(NMAX), Y(NMAX)
INTEGER          ICOL(LA), IDIAG(NMAX), IPIVP(NMAX), IPIVQ(NMAX),
+               IROW(LA), ISTR(NMAX+1), IWORK(LIWORK)
*   .. External Subroutines ..
EXTERNAL         F11DAF, F11DBF
*   .. Executable Statements ..
WRITE (NOUT,*) 'F11DBF Example Program Results'
WRITE (NOUT,*)

*
*   Skip heading in data file
*
READ (NIN,*)

*
*   Read order of matrix and number of non-zero entries
*
READ (NIN,*) N
IF (N.LE.NMAX) THEN
  READ (NIN,*) NNZ
*
*   Read the matrix A
*
DO 20 I = 1, NNZ
  READ (NIN,*) A(I), IROW(I), ICOL(I)
20 CONTINUE
*
*   Read the vector y
*
READ (NIN,*) (Y(I),I=1,N)
*
*   Calculate LU factorization
*
LFILL = -1
DTOL = 0.e0
PSTRAT = 'C'
MILU = 'N'
IFAIL = 0
*
CALL F11DAF(N,NNZ,A,LA,IROW,ICOL,LFILL,DTOL,PSTRAT,MILU,IPIVP,
+         IPIVQ,ISTR,IDIAG,NNZC,NPIVM,IWORK,LIWORK,IFAIL)
*
*   Check value of NPIVM
*

```

```

        IF (NPIVM.GT.0) THEN
*
*       WRITE (NOUT,*) 'Factorization is not complete'
*
        ELSE
*
*       Solve P L D U x = y
*
*       TRANS = 'N'
*       CHECK = 'C'
*
*       CALL F11DBF(TRANS,N,A,LA,IROW,ICOL,IPIVP,IPIVQ,ISTR,IDIAG,
+               CHECK,Y,X,IFAIL)
*
*       Output results
*
*       WRITE (NOUT,*) ' Solution of linear system'
*       DO 40 I = 1, N
*           WRITE (NOUT,'(e16.4)') X(I)
40      CONTINUE
*
*       END IF
*
*       END IF
*       STOP
*       END

```

## 9.2 Program Data

F11DBF Example Program Data

```

4           N
11          NNZ
1.   1   2
1.   1   3
-1.  2   1
2.   2   3
2.   2   4
3.   3   1
-2.  3   4
1.   4   1
-2.  4   2
1.   4   3
1.   4   4          A(I), IROW(I), ICOL(I), I=1,...,NNZ
5.0 13.0 -5.0 4.0  Y(I), I=1,...,N

```

## 9.3 Program Results

F11DBF Example Program Results

```

Solution of linear system
0.1000E+01
0.2000E+01
0.3000E+01
0.4000E+01

```